

```
//*****  
//  
// NAME:          WPCH_TC0.WPM  
//  
// PURPOSE:      Coach on Table of Contents  
//  
// AUTHOR:       Paul Laing, WordPerfect Corp.  
//  
//*****
```

```
// Revision History  
//  
//
```

```
Global(ScrnMode;MsgAdj;DefTitle;CancelMsg)  
Global(SaveScrn;HotElement)
```

```
DefTitle="Coach"           // Main title name  
ScrnMode=?ScreenHigh-24  
CancelMsg=1
```

```
Use("wpchlibr.wpm")       // Shared library file  
ColorScheme()  
ScreenSetup()
```

```
MsgAdj=DlgMessage()
```

```

//*****
//
// MAIN PROGRAM: Coach
//
//*****
Error(Off!)
Cancel(Off!)
Display(Off!)

steps={  "Select and mark headings";
        "Position the table of contents";
        "Generate the table of contents";
      }
var[0]=3
Call(Overview)
If(Result = -1)
  QuitCoach()
Else
  While(1)
    If(var[1])
      Chain("wpchtc01.wpm")
    Endif
    Break
  Endwhile
  While(1)
    If(var[2])
      Chain("wpchtc02.wpm")
    Endif
    Break
  Endwhile
  While(1)
    If(var[3])
      Chain("wpchgn01.wpm")
    Endif
    Break
  Endwhile
Endif
Return

//*****
// END MAIN PROGRAM
//*****

```

```

//*****
//
// SUBROUTINE: Overview
//
//*****
Label(Overview)
  Result=0
  Repeat
    w=65 h=23
    DlgCreate(Result;DefTitle;DlgNoOK!+DlgNoCancel!;;;w;h)
      DlgControl (CtrlPushButton!;;" ~Continue ";
        StyDefault!;w-29;h-2)
      DlgControl (CtrlPushButton!;;"~Quit...Esc";
        StyCancel!;w-15;h-2)

      Message(&tbox;1105)
      ForNext(x;1;tbox[0])
        DlgControl(CtrlLabel!;;tbox[x];;2;x;w-1;1)
      Endfor
      LnHgt=x

      ForNext(x;1;steps[0])
        DlgControl(CtrlRadioButton!;
          var[x];steps[x];;8;x+tbox[0];w-1;1)
      Endfor
      LnHgt=LnHgt+steps[0]

      Message(&lbox;1106)
      ForNext(x;1;lbox[0])
        DlgControl(CtrlLabel!;;lbox[x];;2;x+LnHgt;w-1;1)
      Endfor

      var[1]=1 var[2]=0 var[3]=0
    DlgEnd
  Until(Result <> 0)
Return

```

```

//*****
//
// PROCEDURE: Message
//
//*****
PROCEDURE Message(&tbox;Msg)

Switch(Msg)

CaseOf 1105:
  tbox={ "";
    "      [Bold On]Creating a Table of Contents[Bold Off]";
    "";
    "You can easily create a table of contents for a document.";
    "This table can appear wherever you specify and can be";
    "arranged according to heading levels similar to the";
    "following:";
    "";
    "  Heading Level 1.....1";
    "  Heading Level 2.....2";
    "  Heading Level 3.....3";
    "";
    "There are three steps to creating a Table of Contents.";
    "";
  }

CaseOf 1106:
  tbox={ "Select the step you are interested in, then choose [Bold On]Continue.";
  }

DEFAULT:
tbox={  "No message found!";
      "";
  }

EndSwitch
ENDPROC

```